

## Appendix K Specimen WinBUGS code

All synthesis models performed for this guideline conformed to the principles set out in Appendix D.

### ***Random-effects network meta-analysis for dichotomous data***

```
# Binomial likelihood, logit link
# Random-effects model for multi-arm trials
# based on
# Dias, S., Welton, N.J., Sutton, A.J. & Ades, A.E.
# NICE DSU Technical Support Document 2: A Generalised Linear Modelling Framework
# for Pairwise and Network Meta-Analysis of Randomised Controlled Trials. 2011.
# http://www.nicedsu.org.uk

model {
for(i in 1:NumStudies) {                                # indexes studies
  mu[i] ~ dnorm(0, .0001)                            # vague priors for all trial baselines
  delta[i,1] <- 0                                     # effect is zero for control arm
  w[i,1] <- 0                                         # multi-arm adjustment = zero for ctrl
  for (j in 1:NumArms[i]) {                            # indexes arms
    k[i,j]      ~ dbin(p[i,j],N[i,j])                # binomial likelihood
    logit(p[i,j]) <- mu[i] + delta[i,j]              # model for linear predictor
    rhat[i,j]   <- p[i,j] * N[i,j]                   # expected value of the numerators
    dev[i,j]    <- 2 * (k[i,j] * (log(k[i,j])-log(rhat[i,j])) +
                        + (N[i,j]-k[i,j]) * (log(N[i,j]-k[i,j]) - log(N[i,j]-rhat[i,j]))) # deviance contribution
    dummy[i,j]   <- ArmNo[i,j]                         # data not used in this model
  }
  for (j in 2:NumArms[i]) {                            # indexes arms
    delta[i,j] ~ dnorm(md[i,j],taud[i,j])            # trial-specific LOR distributions
    md[i,j]    <- d[Rx[i,j]] - d[Rx[i,1]] + sw[i,j] # mean of LOR distributions (with
                                                       # multi-arm trial correction)
    taud[i,j]  <- tau *2*(j-1)/j                    # precision of LOR distributions (with
                                                       # multi-arm trial correction)
    w[i,j]     <- (delta[i,j] - d[Rx[i,j]] + d[Rx[i,1]]) # adjustment for multi-arm RCTs
    sw[i,j]    <- sum(w[i,1:j-1])/(j-1)               # cumulative adjustment for multi-arm
                                                       # trials
  }
  resdev[i]    <- sum(dev[i,1:NumArms[i]])           # summed deviance contribution
  dummy2[i]    <- Yrs[i] * RefID[i]                  # data not used in this model
}
totresdev    <- sum(resdev[])                         # total residual deviance

d[1]<-0                                              # effect is 0 for reference treatment
for (j in 2:NumRx) {                                    # indexes treatments
  d[j] ~ dnorm(0, .0001)                            # vague priors for treatment effects
}
sd ~ dunif(0,5)                                       # vague prior for between-trial SD
tau <- pow(sd,-2)                                     # between-trial precision

# Provide estimates of treatment effects T[k] on the natural (probability) scale
AMean ~ dnorm(meanA, precA)
APred ~ dnorm(predA, predPrecA)
for (j in 1:NumRx) {
  logit(Tmean[j]) <- AMean + d[j]
  logit(Tpred[j]) <- APred + d[j]
```

```

}

# pairwise ORs and LORs for all possible pairwise comparisons
for (c in 1:(NumRx-1)) {
  for (j in (c+1):NumRx) {
    lOR[c,j] <- (d[j]-d[c])
    OR[c,j] <- exp(d[j]-d[c])
  }
}

# ranking on relative scale
for (j in 1:NumRx) {
  rk[j] <- blnHiGood*(NumRx+1-rank(d[],j)) + (1-blnHiGood)*rank(d[],j)
  best[j] <- equals(rk[j],1) # probability that treat j is best
  for (h in 1:NumRx) {
    pRk[h,j] <- equals(rk[j],h) # probability that treat j is hth best
  }
}
dummy3 <- YrsA # not used in this model
}

```

## **Random-effects network meta-analysis for continuous data**

```

# Normal likelihood, identity link
# Random-effects model for multi-arm trials
# based on
# Dias, S., Welton, N.J., Sutton, A.J. & Ades, A.E.
# NICE DSU Technical Support Document 2: A Generalised Linear Modelling Framework
# for Pairwise and Network Meta-Analysis of Randomised Controlled Trials. 2011.
# http://www.nicedsu.org.uk

model {
  for(i in 1:NumStudies) { # indexes studies
    w[i,1] <- 0 # multi-arm adjustment = 0 for control
    delta[i,1] <- 0 # treatment effect is 0 for control
    mu[i] ~ dnorm(0, .0001) # vague priors for all trial baselines
    for (j in 1:NumArms[i]) { # indexes arms
      se[i,j] <- SD[i,j] / sqrt(N[i,j])
      var[i,j] <- pow(se[i,j],2) # calculate variances
      prec[i,j] <- 1/var[i,j] # set precisions
      MC[i,j] ~ dnorm(theta[i,j], prec[i,j]) # normal likelihood
      theta[i,j] <- mu[i] + delta[i,j] # model for linear predictor
      dev[i,j] <- (MC[i,j] - theta[i,j]) * (MC[i,j] # deviance contribution
        - theta[i,j]) * prec[i,j]
      dummy[i,j] <- ArmNo[i,j] # data not used in this model
    } # close arm loop
    for (j in 2:NumArms[i]) { # indexes arms
      delta[i,j] ~ dnorm(md[i,j],taud[i,j]) # trial-specific MD distributions
      md[i,j] <- d[Rx[i,j]] - d[Rx[i,1]] + sw[i,j] # mean of MD dists, with multiarm
      taud[i,j] <- tau *2*(j-1)/j # precision of MD dists, with multiarm
      w[i,j] <- (delta[i,j] - d[Rx[i,j]] + d[Rx[i,1]]) # adjustment, multi-arm RCTs
      sw[i,j] <- sum(w[i,1:j-1])/(j-1) # cumulative adjustment for multi-arm
    } # close study loop
    resdev[i] <- sum(dev[i,1:NumArms[i]]) # summed deviance contribution
    dummy2[i] <- Yrs[i] * RefID[i] # data not used in this model
  } # total residual deviance
  totresdev <- sum(resdev[])
}

d[1]<-0 # effect is 0 for reference treatment
for (j in 2:NumRx) { # indexes treatments
  d[j] ~ dnorm(0, .0001) # vague priors for treatment effects

```

```

        }
        # close treatment loop
sd ~ dunif(0,5)           # vague prior for between-trial SD
tau <- pow(sd,-2)          # between-trial precision

# Provide estimates of treatment effects T[j] on the natural (probability) scale
# Given a Mean Effect, meanA, for 'standard' treatment A,
# with precision (1/variance) precA

AMean ~ dnorm(meanA, precA)
APred ~ dnorm(predA, predPrecA)
for (j in 1:NumRx) {
  Tmean[j] <- AMean + d[j]
  Tpred[j] <- APred + d[j]
}

# pairwise MDs for all possible pairwise comparisons
for (c in 1:(NumRx-1)) {
  for (j in (c+1):NumRx) {
    MD[c,j] <- (d[j] - d[c])
  }
}

# ranking on relative scale
for (j in 1:NumRx) {
  rk[j]      <- blnHiGood*(NumRx+1-rank(d[],j)) + (1-blnHiGood)*rank(d[],j)
  best[j]    <- equals(rk[j],1)                                # probability that treat j is best
  for (h in 1:NumRx) {
    pRk[h,j]  <- equals(rk[j],h)                            # probability that treat j is hth best
  }
}
dummy3      <- YrsA                                         # data not used in this model
}

```

## **Random-effects network meta-analysis for categorical data**

```

# Binomial likelihood, probit link (different categories)
# Random-effects model for multi-arm trials
# based on
# Dias, S., Welton, N.J., Sutton, A.J. & Ades, A.E.
# NICE DSU Technical Support Document 2: A Generalised Linear Modelling Framework
# for Pairwise and Network Meta-Analysis of Randomised Controlled Trials. 2011.
# http://www.nicedsu.org.uk

model {
  for(i in 1:numStudies) {                               # indexes studies
    w[i,1]      <- 0                                  # multi-arm adjustment =0 for control arm
    delta[i,1]   <- 0                                  # treatment effect =0 for control arm
    mu[i]       ~ dnorm(0, .0001)                      # vague priors for all trial baselines
    for (j in 1:numArms[i]) {                           # indexes arms
      p[i,j,1]   <- 1                                 # Pr(score>0 = 1)
      for (c in 1:numCats[i]-1) {                      # indexes categories
        n[i,j,c]   <- N[i,j] - sum(k[i,j,1:c]) + k[i,j,c] # calculate ns for each consecutive category
        k[i,j,c]   ~ dbin(q[i,j,c],n[i,j,c]) # binomial likelihood
        q[i,j,c]   <- 1-(p[i,j,C[i,c+1]]/p[i,j,C[i,c]]) # conditional probabilities
        theta[i,j,c] <- mu[i] + delta[i,j] + z[C[i,c+1]-1] # linear predictor
        rhat[i,j,c] <- q[i,j,c] * n[i,j,c] # predicted number events
        dv[i,j,c]   <- 2 * (k[i,j,c]*(log(k[i,j,c]) - log(rhat[i,j,c])) # deviance
        +(n[i,j,c] - k[i,j,c]) * (log(n[i,j,c]-k[i,j,c]))
      }
    }
  }
}

```

```

        - log(n[i,j,c] - rhat[i,j,c]))
                                # deviance contribution
    }
                                # close category loop
for (c in 2:numCats[i]) {
                                # indexes categories again
    p[i,j,C[i,c]]      <- 1 - phi(theta.adj[i,j,c])
                                # link function
    theta.adj[i,j,c]    <- step(-5-theta[i,j,c-1]) * -5
                        + step(theta[i,j,c-1]-5) * 5
                        + step(5-theta[i,j,c-1]) * step(theta[i,j,c-1]+5)
                        * theta[i,j,c-1]
                                # constrain theta to [-5,5] to avoid
                                # numerical errors when using phi()
}
                                # close second category loop
dev[i,j] <- sum(dv[i,j,1:numCats[i]-1])
dummy[i,j] <- ArmNo[i,j]
}
                                # close arm loop
for (j in 2:numArms[i]) {
                                # indexes arms
    delta[i,j] ~ dnorm(md[i,j], taud[i,j])
    md[i,j]     <- d[Rx[i,j]] - d[Rx[i,1]] + sw[i,j]
                                # mean of LHR distributions, with multi-arm
                                # trial correction
    taud[i,j]   <- tau * 2 * (j-1) / j
                                # precision of LHR distributions, with multi-arm
                                # trial correction
    w[i,j]      <- (delta[i,j] - d[Rx[i,j]] + d[Rx[i,1]])
                                # adjustment, multi-arm RCTs
    sw[i,j]     <- sum(w[i,1:j-1]) / (j-1)
                                # cumulative adjustment for multi-arm trials
}
resdev[i] <- sum(dev[i,1:numArms[i]])
dummy2[i] <- RefID[i] + juncture[i]
}
                                # close study loop

z[1] <- 0
for (j in 2:maxNumCats-1) {
    zz[j] ~ dunif(0,5)
    z[j]  <- z[j-1] + zz[j]
}
totresdev <- sum(resdev[])
                                # total residual deviance

d[1] <- 0
for (j in 2:numRx) {
    d[j] ~ dnorm(0, .0001)
}
sd ~ dunif(0, 5)
tau <- pow(sd, -2)
                                # effect is zero for reference treatment
                                # indexes treatments
                                # vague priors for treatment effects
                                # vague prior for between-trial SD
                                # between-trial precision

# z-scores for all possible pairwise comparisons
for (c in 1:(numRx-1)) {
    for (j in (c+1):numRx) {
        zscore[c,j]     <- d[j] - d[c]
    }
}

# Provide estimates of treatment effects T[k] on the natural (probability) scale
# at each juncture
for (j in 1:numJunctures) {
    AMean[j] ~ dnorm(meanA[j], precA[j])
    APred[j] ~ dnorm(predA[j], predPrecA[j])
    for (i in 1:numRx) {
        # calculate prob of achieving each level
        for (c in 1:maxNumCats-1) {
            Tmean[j,i,c] <- 1 - phi(AMean[j] + d[i] + z[c])
            Tpred[j,i,c] <- 1 - phi(APred[j] + d[i] + z[c])
        }
    }
}

```

```

        }
    }
}

# ranking on relative scale
for (j in 1:numRx) {
    rk[j]      <- blnHiGood * (numRx+1 - rank(d[],j)) + (1-blnHiGood) * rank(d[],j)
    best[j]    <- equals(rk[j],1)                                # probability that treat j is best
    for (h in 1:numRx) {
        pRk[h,j]  <- equals(rk[j],h)                            # probability that treat j is hth best
    }
}
}

```

## **Random-effects network meta-analysis for categorical data, with adjustment for dose**

```

# Binomial likelihood, probit link (different categories)
# Random-effects model for multi-arm trials
# based on
# Dias, S., Welton, N.J., Sutton, A.J. & Ades, A.E.
# NICE DSU Technical Support Document 2: A Generalised Linear Modelling Framework
# for Pairwise and Network Meta-Analysis of Randomised Controlled Trials. 2011.
# and
# Dias, S., Sutton, A.J., Welton, N.J. & Ades, A.E.
# NICE DSU Technical Support Document 3: Heterogeneity: subgroups, meta-regression,
# bias and bias-adjustment. 2011.
#
# http://www.nicedsu.org.uk

model {
for(i in 1:numStudies) {                                # indexes studies
    w[i,1]      <- 0                                  # multi-arm adjustment =0 for control arm
    delta[i,1]   <- 0                                  # treatment effect =0 for control arm
    mu[i]        ~ dnorm(0, 0.01)                      # vague priors for all trial baselines
    for (j in 1:numArms[i]) {                          # indexes arms
        p[i,j,1] <- 1                                 # Pr(score>0 = 1)
        X.CDose[i,j] <- (X.Dose[RefID[i],ArmNo[i,j]] - avgD[Rx[i,j]]) / 1000
                                                # centre dose covariate and divide by
                                                # 1000 (which calms WinBUGS down)
        BX[i,j] <- B[Rx[i,j]] * X.CDose[i,j] - B[Rx[i,1]] * X.CDose[i,1]
                                                # multiply centred dose covariate for arm and
                                                # 'control' by treatment-specific
                                                # dose--response coefficients
    for (c in 1:numCats[i]-1) {                        # indexes categories
        n[i,j,c]     <- N[i,j] - sum(k[i,j,1:c]) + k[i,j,c]
                                                # calculate ns for each consecutive category
        k[i,j,c]     ~ dbin(q[i,j,c], n[i,j,c]) # binomial likelihood
        q[i,j,c]     <- 1-(p[i,j,C[i,c+1]]/p[i,j,C[i,c]])
                                                # conditional probabilities
        theta[i,j,c] <- mu[i] + delta[i,j] + z[C[i,c+1]-1] + BX[i,j]
                                                # linear predictor
        rhat[i,j,c]  <- q[i,j,c] * n[i,j,c]      # predicted number events
        dv[i,j,c]    <- 2 * (k[i,j,c]*(log(k[i,j,c]) - log(rhat[i,j,c]))
                                +(n[i,j,c] - k[i,j,c]) * (log(n[i,j,c]-k[i,j,c])
                                - log(n[i,j,c] - rhat[i,j,c])))
                                                # deviance contribution
    }
    dev[i,j] <- sum(dv[i,j,1:numCats[i]-1])      # deviance contribution of each arm
    for (c in 2:numCats[i]) {                        # indexes categories

```

```

p[i,j,C[i,c]] <- 1 - phi.adj[i,j,c]      # link function
phi.adj[i,j,c] <- step(5+theta[i,j,c-1])
  * (step(theta[i,j,c-1]-5)
  + step(5-theta[i,j,c-1]) * phi(theta[i,j,c-1]))
  # adjust phi(x) for values that can give
  # numerical errors when x< -5, phi(x)=0,
  # when x> 5, phi(x)=1
}
dummy[i,j] <- X.FixedDose[RefID[i],ArmNo[i,j]]  # data not used in this model
}
for (j in 2:numArms[i]) {                      # indexes arms
  delta[i,j] ~ dnorm(md[i,j], taud[i,j])
  md[i,j]    <- d[Rx[i,j]] - d[Rx[i,1]] + sw[i,j]
  taud[i,j]   <- tau * 2 * (j-1) / j          # mean of LHR distributions, with multi-arm
                                                # trial correction
  # precision of LHR distributions, with multi-
  # arm trial correction
  w[i,j]     <- (delta[i,j] - d[Rx[i,j]] + d[Rx[i,1]]) # adjustment, multi-arm RCTs
  sw[i,j]    <- sum(w[i,1:j-1]) / (j-1)        # cumulative adjustment for multi-arm trials
}
dummy2[i] <- juncture[i]
resdev[i] <- sum(dev[i,1:numArms[i]])         # summed residual deviance for this trial
}
sd ~ dunif(0, 2)                                # vague prior for between-trial SD
tau <- pow(sd, -2)                               # between-trial precision

z[1] <- 0                                         # set z=0 for bottom category
for (j in 2:maxNumCats-1) {
  zz[j] ~ dunif(0,5)                            # prior for difference between z[j] and z[j-1]
  z[j]  <- z[j-1] + zz[j]                         # prior for z[j]
}
totresdev <- sum(resdev[])                        # total residual deviance

# calculate mean dose (for centring covariates)
for (t in 1:numRx) {                             # indexes treatments
  for (s in 1:numStudies) {                      # indexes studies
    for (a in 1:numArms[s]) {                    # indexes arms
      dd[s,a,t] <- X.Dose[RefID[s],ArmNo[s,a]] * equals(Rx[s,a], t)
      # record doses where treatment in arm a is t
      nn[s,a,t] <- equals(Rx[s,a], t)           # record a 1 where treatment in arm a is t
    }
    ddd[t,s] <- sum(dd[s,1:numArms[s],t])       # sum doses for treatment t for study s
    nnn[t,s] <- sum(nn[s,1:numArms[s],t])        # count arms with treatment t for study s
  }
  avgD[t] <- sum(ddd[t,1:numStudies]) / sum(nnn[t,1:numStudies])
  # sum doses for all study-arms, divide by
  # no. of arms to calculate average dose
}
# close treatment loop

d[1] <- 0                                         # effect is zero for placebo
B[1] <- 0                                         # dose--response coefficient is 0 for placebo
bSD <- 2                                           # moderately vague SD for dose--response prior
prec <- pow(bSD, -2)                              # convert SD to precision
for (j in 2:numRx) {
  d[j] ~ dnorm(0, .0001)                          # vague priors for treatment effects
  B[j] ~ dnorm(0, prec)                           # priors for dose--response coefficients
}
# close treatment loop

# fit basic parameters (relative effects) at specified covariate values

```

```

for (t in 1:numRx) {
    dz[t] <- d[t] + X.FitVal[t] / 1000 * B[t]
} # close treatment loop

# pairwise z-scores for all possible comparisons
for (c in 1:(numRx-1)) {
    for (j in (c+1):numRx) {
        zscore[c,j] <- dz[j] - dz[c]
    }
} # close treatment loop

# ranking on relative scale
for (j in 1:numRx) {
    rk[j] <- blnHiGood * (numRx+1 - rank(dz[],j)) + (1-blnHiGood) * rank(dz[],j)
} # compute ranks (invert if necessary)
best[j] <- equals(rk[j],1)
for (h in 1:numRx) {
    pRk[h,j] <- equals(rk[j],h)
}
} # close rank loop
} # close treatment loop

# Provide estimates of treatment effects T[k] on the natural (probability) scale
# at each juncture
for (j in 1:numJunctures) {
    AMean[j] ~ dnorm(meanA[j], precA[j])
    APred[j] ~ dnorm(predA[j], predPrecA[j])
    for (i in 1:numRx) {
        # calculate prob of achieving each level
        for (c in 1:maxNumCats-1) {
            Tmean[j,i,c] <- 1 - phi(AMean[j] + dz[i] + z[c])
            Tpred[j,i,c] <- 1 - phi(APred[j] + dz[i] + z[c])
        }
    }
}
}

```